



# Bluespec Training

## Eg03: Concurrent Bubblesort

## A simple concurrent Bubblesort example.

```
Export FPC@4:
typedef struct { int } cont_t;

module ex_fst_cont2_fstExpType:


Integer file_depth = 15;

function cont2()  determine_group(cont_t val);
return {val};
endfunction


FPC@4@cont_t; intbound0;
return FPC@4@file_depth; the_intbound0(intbound0);
FPC@4@cont_t; outbound0;
return FPC@4@file_depth; the_outbound0(intbound0);
FPC@4@cont_t; outbound0;
return FPC@4@file_depth; the_outbound0(outbound0);

rule eval (True):
  cont_t b_data = intbound0(100);
  FPC@4@cont_t out_group =
    determine_group(b_data
    ) == 0 ? outbound0; outbound0;
  intbound0(1);
endrule; eval;

endmodule; | ex_fst_cont2.bc
```



[www.bluespec.com](http://www.bluespec.com)



# Generalization via 5 versions

The accompanying code demonstrates several versions:

Eg03a_Bubblesort/	Sorts 5 items, each of type <code>`Int 32'</code> . Completely sequential, to show correspondence with conventional software implementation.
Eg03b_Bubblesort/	Parallel version. Uses <code>'maxBound'</code> (largest <code>Int 32</code> ) as a “special” value different from any of the sorted values (like <code>“+infinity”</code> ).
Eg03c_Bubblesort/	Generalizes <code>'5'</code> items to <code>'n'</code> items.
Eg03d_Bubblesort/	Generalizes items of type <code>'Int 32'</code> to items of arbitrary type <code>'t'</code> (i.e., makes the program polymorphic).
Eg03e_Bubblesort/	Uses the <code>'Maybe t'</code> type to eliminate the need for a <code>'maxBound'</code> special value.

Note: this is a pedagogical introductory example focusing on concurrency and modularity, and is not intended as an example of efficient sorting!  
(See `'Eg06_Mergesort'` for more efficient and scalable sorting.)

# 1<sup>st</sup> version: directory Eg03a\_Bubblesort/

Examine the two source files: src/Top.bs src/Bubblesort.bs

*We suggest that you take multiple passes through the files, incrementally increasing your understanding. Initially, just try to understand the syntactic structure, making frequent reference to the lecture slides in “Lec\_Basic\_Syntax” in the “Reference” directory.*

The following excerpts show an outline of the syntactic structure of the two source files.

```
package Top where
...
import ...
import Bubblesort
...
n :: Int 32 = 5
...
mkTop :: Module Empty
mkTop =
  module
    rg_j1 :: Reg (Int 32) <- mkReg 0
    ...
    sorter :: Sort_IFC <- mkBubblesort
    ...
  rules
```

```
package Bubblesort where
...
import ...
...
interface Sort_IFC =
  ...
mkBubblesort :: Module Sort_IFC
mkBubblesort =
  module
    rg_j :: Reg (UInt 3) <- mkReg 0
    ...
  rules
  ...
  method definitions
```

# 1<sup>st</sup> version: directory Eg03a\_Bubblesort/ (contd.)

Examine the two source files: src/Top.bs src/Bubblesort.bs

```
package Top where
...
import ...
import Bubblesort
...
n :: Int 32 = 5
...
mkTop :: Module Empty
mkTop =
  module
    rg_j1 :: Reg (Int 32) <- mkReg 0
    ...
    sorter :: Sort_IFC <- mkBubblesort
    ...
  rules
```


```
package Bubblesort where
...
import ...
...
interface Sort_IFC =
  ...
mkBubblesort :: Module Sort_IFC
mkBubblesort =
  module
    rg_j :: Reg (UInt 3) <- mkReg 0
    ...
  rules
  ...
  method definitions
```

Notes on the syntactic structure:

- The top-level module is mkTop; it has an Empty interface (an interface with no methods).
- It instantiates several sub-modules: a few registers, a pseudo-random number generator (LFSR, for Linear Feedback Shift Register), and the mkBubblesort module.
- The mkBubblesort module, in turn, instantiates some registers as its sub-modules.
- The rules in the Top invoke the put and get methods in the bubblesort module's interface.
- All the rules and methods have Boolean conditions indicating when they are “ready”.

In module mkBubblesort:

- The register rg\_pc is used to sequence the sorting actions. The name is suggestive of “Program Counter”. 3 bits are enough to distinguish all the states.
- The register rg\_j is used to count 5 incoming values. Thus, it is typed UInt 3, i.e., an unsigned integer of 3 bits, capable of holding values 0..7. It’s reset value (initial value) is 0.
- The register rg\_swapped remembers whether any swap occurred in the current pass.
- The registers x0..x4 hold the 5 values to be sorted. Each value is of type Int 32, i.e., a signed integer of 32 bits. Each register’s reset value is maxBound, a symbolic name representing the largest Int 32, i.e.,  $+2^{31-1}$ . We assume all input values to be sorted will be strictly  $< \text{maxBound}$



The module mkBubblesort, showing the registers in isolation

The rules in module mkBubblesort essentially encode a sequential algorithm similar to that shown below in a pseudo-C notation:

```
while (True)
  swapped = False;
  for (pc = 1; pc < 5; pc++) {
    if (x[pc-1] > x[pc] {
      swap them;
      swapped = True;
    }
  }
  if (! swapped) break;
}
```

Each rule has:

- A condition that says when it can fire, and
- A body that says what happens if it fires.

# Building and running the codes

Each variation is built and run in the same way:

- In the Build/ directory you can use the 'Makefile' for building and running Bluesim or Verilog sim:

```
$ make b_compile b_link b_sim // for Bluesim
$ make v_compile v_link v_sim // for Verilog sim
```


Note: When building for Bluesim, compiler-temporary files are created in the build\_bsim/ directory. When building for Verilog, temporaries are in build\_v/ and Verilog files are in verilog\_dir/. 'make clean' and 'make full\_clean' will clean up these directories.

# Build and run the 1<sup>st</sup> version


- In the Build/ directory, practice building and running the program in both ways:
  - Using 'make' commands, for Bluesim.
  - Using 'make' commands, for Verilog sim.
- Observe the inputs and outputs and verify that they are reasonable (there are 5 inputs and outputs, and the outputs are a sorted version of the inputs).
- When you run your simulation, a file "dump.vcd" is created, containing waveforms, which you can view in your favorite waveform viewer.
  - (It is created because of the -V flag for Bluesim or the +bscvcd flag for Verilog sim.)
  - The picture "Waves\_screenshot.tiff" is a screenshot of such a view.
  - Study the waveform and make sure you understand how it reflects the behavior of the Bluespec code.



# Waveforms from the 1<sup>st</sup> version



# Eg03: Basic concurrency and modularity



## Algorithm/Architecture for remaining versions:


- The module accepts a stream of 'n' input items (unsorted).
- These are shifted in to 'n' registers.
- Concurrently (and even while inputs are arriving), whenever two adjacent registers contain values in the wrong order, we swap their contents.
- When 'n' inputs have been received, and all of them are sorted, the module yields a stream of 'n' sorted outputs by shifting them out.
- When 'n' outputs have been streamed out, the module is ready for its next set of 'n' inputs.

In module mkBubblesort:

- The Bool variable “done” defines the condition that all values have been received and are sorted:

```
-- Test if array is sorted
let done :: Bool
    done = ((rg_inj == 5) && (x0 <= x1) && (x1 <= x2) && (x2 <= x3) && (x3 <= x4))
```

- In Bluespec, value-expressions like this just represent combinational circuits:




## 2<sup>nd</sup> version: directory Eg03b\_Bubblesort/ (contd.)

```
-- Inputs: feed input values into x4
put :: Int 32 -> Action
put x = do
    x4 := x
    rg_inj := rg_inj + 1
    when ((rg_inj < 5) && (x4 == maxBound))
```

In module mkBubblesort:

- The “put” method is of type Action, i.e., its body is an expression of type Action.
  - The body has two sub-Actions: one places a new input value x into register x4, and the other increments the input count rg\_j.
- The method condition ensures two things:
  - “(rg\_j < 5)” ensures that we stop after receiving 5 inputs
  - “(x4==maxBound)” ensures that we don’t over-write a meaningful value already in x4, i.e., it can place a new value in x4 only after the swap rules have moved the previous value out and replaced it with maxBound.

Hardware for put method, in isolation




```
-- Outputs: drain by shifting them out of x0
get :: ActionValue (Int 32)
get = do
    x0 := x1
    x1 := x2
    x2 := x3
    x3 := x4
    x4 := maxBound
    if1 (x1 == maxBound) (rg_inj := 0)
    return x0
  when done
```

In module mkBubblesort:

- The “get” method is of type ActionValue (Int 32), i.e., it’s body is an Action, and it also returns a value of type Int 32.
- The body returns x0, shifts all remaining values in x1..x4 down into x0..x3, respectively, and shifts the value maxBound into x4.
- The method condition of the “get” method:
  - Is only enabled when “done” is true, i.e., all inputs have been received and all values are sorted
  - Is only enabled until we’ve returned the 5<sup>th</sup> value, after which x0 contains maxBound
- When we return the 5<sup>th</sup> value (from x0), x1 will be maxBound; at this time we can reset rg\_j to 0
- Note that after returning 5 values, the module is again in it’s original state—rg\_j contains 0, and x0..x4 contain maxBound—and so it is ready to receive the next 5 values to be sorted

```
-- Outputs: drain by shifting them out of x0
get :: ActionValue (Int 32)
get = do
    x0 := x1
    x1 := x2
    x2 := x3
    x3 := x4
    x4 := maxBound
    if1 (x1 == maxBound) (rg_inj := 0)
    return x0
when done
```

## Hardware for “get” method, in isolation



```
"rl_swap_0_1": when (x0 > x1) ==> do
  x0 := x1
  x1 := x0
```

```
"rl_swap_1_2": when (x1 > x2) ==> do
  x1 := x2
  x2 := x1
```

```
"rl_swap_2_3": when (x2 > x3) ==> do
  x2 := x3
  x3 := x2
```

```
"rl_swap_3_4": when (x3 > x4) ==> do
  x3 := x4
  x4 := x3
```

In module mkBubblesort:

- In each of the four rules:
  - The rule condition (an expression of type Bool), e.g., “(x0>x1)” tests if two adjacent values are in the wrong order. This is a prerequisite for the rule to “fire”, i.e., to execute its body.
  - The rule body (an expression of type Action) is composed of two sub-Actions whose effect is to swap the contents of the corresponding two registers. Note:
    - Expressions/methods/functions of type “Action” or “ActionValue t” (potentially) change the state of the system (e.g., write to a register, write to memory, enqueue onto a FIFO, etc.). We also say that such expressions have “side-effects”.
    - Expressions that are not of type Action or ActionValue (e.g., Bool) *never* change the state of the system (this is guaranteed by Bluespec’s type-checking rules). We also say that these are “pure” or “value” expressions.
    - All Actions in a rule are considered instantaneous and simultaneous. The two assignments are not read sequentially (as is typical in software programs), and this explains why there is no need for a “temporary” intermediate variable (as is typical in software programs)


```
"rl_swap_0_1": when (x0 > x1) ==> do  
  x0 := x1  
  x1 := x0
```

```
"rl_swap_1_2": when (x1 > x2) ==> do  
  x1 := x2  
  x2 := x1
```


```
"rl_swap_2_3": when (x2 > x3) ==> do  
  x2 := x3  
  x3 := x2
```

```
"rl_swap_3_4": when (x3 > x4) ==> do  
  x3 := x4  
  x4 := x3
```

Hardware for rl\_swap\_1\_2, in isolation



... and in more detail





In module mkTop:

- rg\_j1 and rg\_j2 are used to count 5 inputs and 5 outputs respectively. They are somewhat arbitrarily typed at Int 32 (since they count 0..4, even Int 3 or UInt 3 would have been ok, but we usually don't worry so much about minimizing state bits in a testbench).
- mkLFSR\_8 instantiates, from the Bluespec library, a pseudo-random number generator of 8-bit values (see Sec.C.8.1 in the Reference Guide).
- In rule "rl\_feed\_inputs":
  - The rule condition ensures that is only enabled to generate 5 inputs
  - lfsr.value() is a "value method" that returns a Bit 8; zeroExtend() extends this to Bit 32; unpack() converts this to an Int 32. lfsr.next() is an Action method that tells the LFSR to advance to its next random number
  - "sorter.put(x)" sends this new input into the bubblesort module
    - Recall that the "put" method in mkBubblesort has a method condition that prevents overwriting a previous value. Thus, rl\_feed\_inputs can only fire when that condition is true.
- In rule "rl\_drain\_outputs"
  - The rule condition ensures that it only retrieves 5 outputs
  - "sorter.get()" yields the next output, whenever the method condition on the "get" method allows, and this value is displayed.

### Summary of behavior:

- Rule “rl\_feed\_inputs” in the testbench module feeds 5 random values into the sorter module using the “put” method.
- The sorter module receives the 5 inputs via the “put” method. Sorting begins as soon as the first input arrives (the first step sorts the first input against its +infinity neighbor), and continues as more inputs arrive. Sorting may continue for some time after all inputs arrive, until all 5 registers are sorted.
- When all 5 registers are sorted, the “get” method is enabled, and the testbench drains out the 5 values in order. The sorter module yields these 5 values in order by shifting them through the registers.
- After yielding 5 outputs, the sorter module is back in its initial state, and ready for another 5 inputs (although our testbench quits after the first set).

*This behavior can be seen in the waveforms for this design created during simulation. We will shortly describe simulation and waveform generation but, for your convenience, the directory includes a screen shot (“waves\_screenshot.tiff”) which you can view directly.*

# Build and run the 2<sup>nd</sup> version

- In the Build/ directory, practice building and running the program in both ways:
  - Using 'make' commands, for Bluesim.
  - Using 'make' commands, for Verilog sim.
- Observe the inputs and outputs and verify that they are reasonable (there are 5 inputs and outputs, and the outputs are a sorted version of the inputs).
- Note that an input is supplied only on *every other* clock. Why?
  - (Hint: see previous remarks on method condition of the "put" method.)
- When you run your simulation, a file "dump.vcd" is created, containing waveforms, which you can view in your favorite waveform viewer.
  - (It is created because of the -V flag for Bluesim or the +bscvcd flag for Verilog sim.)
- The picture "waves\_screenshot.tiff" is a screenshot of such a view. Note that it shows Bluespec source code types Int 32, Bool, ..., and not raw Verilog bits.
- Study the waveform and make sure you understand how it reflects the behavior of the Bluespec code.

# Suggested exercises

*In this and future examples, we suggest extra exercises to deepen your understanding of Bluespec, which you can pursue on your own (may not be covered during classroom training)*

- Examine the generated Verilog file mkBubblesort.v (in the “verilog\_RTL/” directory).
  - Look at the input and output ports, and understand how they correspond to the Bluespec interface Sort\_IFC and its methods.
  - Skim the interior of the Verilog module, and notice correspondences with the Bluespec source module (registers, rules, rule and method conditions, ...).
- Analyze and explain in detail what would happen if the testbench supplied ‘maxBound’ as one (or more) of the input values. If you wish, modify the testbench to test this.
- Modify the testbench to sort two (or three, or more) rounds of 5-input sequences, instead of quitting after the first round.
- Modify the testbench and the bubblesort modules to sort sequences of some other length instead of 5 (say, 10).
- When you compile, you'll see messages that a rule was treated as “more urgent” than another.
  - Look at the Makefile, and you'll see that by default it's compiled with OPTION1
  - Try it with OPTION2, OPTION3 and OPTION4, and understand the code for them

# Time-out to reinforce some concepts

Before moving on with the examples, please study the lecture: [Lec\\_Rule\\_Semantics](#) to understand the concepts behind Rules:

- Semantics of individual rules (and the methods they call)
  - “simultaneous/parallel actions”, “instantaneous”
- Semantics of concurrency of a set of rules in each clock
  - “concurrency”
  - “ordering constraints” between methods in different rules
  - “rule schedules”

## 3<sup>rd</sup> version: directory Eg03c\_Bubblesort/

The 3<sup>rd</sup> version generalizes the 2<sup>nd</sup> version from sorting 5 numbers to sorting  $n$  numbers

Examine the two source files: src/Top.bs src/Bubblesort.bs

It is useful to compare these side-by-side with the corresponding two files in Eg03b\_Bubblesort/src/, to see how they have changed.

Notice that the interface type is now parameterized by  $n$ :

```
interface (Sort_IFC :: # -> *) n_t =
```

In Bluespec, certain types and type parameters can be *numeric types* (of “kind” #).

Note that *numeric types* (which are only meaningful at compile time) are distinct from *numeric values* (which can of course occur at run time).

Bluespec is very strict in type-checking—there are no automatic conversions

The code uses these explicit conversions:



The 2<sup>nd</sup> version's explicit registers x0..x4 have here been replaced by a vector of registers:

```
-- A vector of registers to hold the values being sorted
-- Note: 'maxBound' is largest 'Int 32'; we assume none of the
-- actual values to be sorted have this value.
xs :: Vector n_t (Reg (Int 32)) <- replicateM (mkReg maxBound)
```


“replicateM” repeatedly applies “mkReg(maxBound)” to create them.

# 3<sup>rd</sup> version: directory Eg03c\_Bubblesort/ (contd.)

The 2<sup>nd</sup> version's explicit “done” test is here computed using List.all and a lambda-expression predicate:

```
-- Test if array is sorted
let done :: Bool
    done = ( (rg_inj == fromInteger n)
              && (List.all (\i -> ((xs !! i)._read <= (xs !! (i+1))._read))
                (List.upto 0 (n - 2))))
```

Note: this List.all “loop” is also “statically elaborated” (i.e., unfolded by the compiler) into a circuit of combinational logic



Observe that, unlike traditional languages, there is no storage location corresponding to loop variables like *i* and the accumulated value (call it “*b*”). The former has disappeared completely, and the latter exists in multiple versions (*b*, *b'*, *b''*, ...) each of which just names a wire.



The 2<sup>nd</sup> version's separately-written rules are here replaced by a List.map loop generating them:

```
-- Function to generate rule to swap xs[i] and xs[i+1] if unordered
gen_swap_rule :: Integer -> Rules
gen_swap_rule i = let
    xs_i      = xs !! i
    xs_i_plus_1 = xs !! (i+1)
  in
    rules
      when (xs_i > xs_i_plus_1) ==> do
        xs_i      := xs_i_plus_1
        xs_i_plus_1 := xs_i

-- Add the rules to the module in descending urgency order
addRules_descending_urgency (List.map gen_swap_rule (List.upto 0 (n - 2)))
```

Note: this for-loop is “statically elaborated” (i.e., unfolded by the compiler), and is equivalent to writing out n-1 rules explicitly.

In the “get” method, the following line:

```
writeVReg xs (shiftInAtN (readVReg xs) maxBound)
```

uses a number of Bluespec library functions on vectors:

- readVReg() returns a vector of values corresponding to the contents of a vector of registers
- shiftInAtN() takes a vector of values  $x_1, \dots, x_N$  and a new value  $y$ , and returns a vector of values  $x_2, \dots, x_N, y$  (discards  $x_1$ )
- writeVReg() writes a vector of values into a vector of registers

Note: this “loop” is also “statically elaborated” (i.e., unfolded by the compiler)

# Time-out to reinforce some concepts

Before moving on with the examples, please study the lecture: `Lec_Types` to understand the concepts behind types, polymorphism, and numeric types.

Please also *skim* through Section C.3 (“Vectors”) in the Reference Guide.

# Synthesis hierarchy

- In the 1<sup>st</sup> and 2<sup>nd</sup> versions, in front of 'module mkBubblesort' line, there is a {-# verilog...#-} attribute:

```
{-# verilog mkBubblesort #-}  
  
mkBubblesort :: Module Sort_IFC  
mkBubblesort =  
    module
```

- When generating Verilog, this creates a 'mkBubblesort.v' file with a 'mkBubblesort' Verilog module
- In the 3<sup>rd</sup> version, this {-# verilog...#-} attribute is removed.
- This is because Bluespec cannot separately synthesize *polymorphic* modules; they can only be inlined into a parent module
- Instead, in Top.bs, we have created a specific instance of the module (for  $n=20$ ); since this is no longer polymorphic, it can be separately synthesized. This is the module actually instantiated in the module mkTop:

```
-- Instantiate and separately synthesize a Bubblesort module for size 'N_t'  
  
{-# verilog mkBubblesort_nt #-}  
  
mkBubblesort_nt :: Module (Sort_IFC N_t)  
mkBubblesort_nt =  
    module  
        m :: Sort_IFC N_t <- mkBubblesort  
        return m
```

- This is a common idiom in Bluespec, for creating a separately synthesized instance of a polymorphic module

# Build and run the 3<sup>rd</sup> version

- In the Build directory, build and run using the 'make' commands, with Bluesim and/or with Verilog sim, as described earlier
- Observe the inputs and outputs and verify that they are reasonable (that there are 20 inputs and outputs, and the outputs are a sorted version of the inputs)
- In Top.bs, change the 1 line that defines the size of the problem:

```
type N_t = 20
```

Rebuild and re-run to test it.

## 4<sup>th</sup> version: directory Eg03d\_Bubblesort/

The 4<sup>th</sup> version generalizes the 3<sup>rd</sup> version from sorting values of type `Int 32` to sorting values of any type “`t`”

Examine the two source files: `src/Top.bs` `src/Bubblesort.bs`

It is useful to compare these side-by-side with the corresponding two files in the previous version, to see how they have changed.

Notice that the interface type is now further parameterized by `t`:

```
interface (Bubblesort_IFC :: # -> * -> *) n_t t =
```

The `mkBubblesort` type declaration is now extended with *type contexts* (we also call them *provisos*):

```
mkBubblesort :: (Bits t wt,           -- ensures 't' has a hardware bit representation
                 Ord t,              -- ensures 't' has the '<=' comparison operator
                 Eq t,               -- ensures 't' has the '==' comparison operator
                 Bounded t)         -- ensures 't' has a 'maxBound' element
=>
Module (Bubblesort_IFC n_t t)
```

These are assertions, respectively, that:

- values of type `t` have a bit representation (with bit-width `wt`), since we need to store them in registers
- values of type `t` can be compared with the `<=` operator (ordering)
- values of type `t` can be compared with the `==` operator (equality)
- there exists a ‘`maxBound`’ value of type `t`

Without these assertions, the compiler would emit a type-checking error, since all these properties are used inside the module.

# Time-out to reinforce some concepts

Please study the lecture: `Lec_Typeclasses`  
to understand the concepts behind `Bits`, `Ord`, `Eq`, `Bounded`.

# Build and run the 4<sup>th</sup> version

- In the Build directory, build and run using the 'make' commands, with Bluesim and/or with Verilog sim, as described earlier
- Observe the inputs and outputs and verify that they are reasonable (that there are 20 inputs and outputs, and the outputs are a sorted version of the inputs)
- In Top.bs, change the two lines that define the size of the problem and the type of values being sorted:

```
type N_t = 20  
type MyT = UInt 24
```

Rebuild and re-run to test it.



## 5<sup>th</sup> version: directory Eg03e\_Bubblesort/

The 5<sup>th</sup> version eliminates the dependency on the existence of a separate ‘maxBound’ value that is separate from legitimate input values.

Before proceeding, please review the lecture: Lec\_Types  
and, specifically, the section on “Maybe” types.

Examine the two source files: src/Top.bs src/Bubblesort.bs

It is useful to compare these side-by-side with the corresponding two files in the previous version, to see how they have changed.

Notice that the “Bounded t” proviso has been removed from module mkBubblesort.

The vector of registers now contain values of type “Maybe t” instead of “t”:

```
-- A vector of registers to hold the values being sorted
xs :: Vector n_t (Reg (Maybe t)) <- replicateM (mkReg Invalid)
```

These are initialized/reset to the value “Invalid”.

This plays the role previously played by “maxBound”.

In the “put” method, instead of inserting x into the register, we insert:

```
xs !! jMax := Valid x
```

After the module mkBubblesort, there is some new code

```
instance (Ord t) => Ord (Maybe t) where
  (<=) :: (Maybe t) -> (Maybe t) -> Bool
  mx1 <= mx2 = case (mx1, mx2) of
    ...
```

A value of type Maybe t has two forms:

- Invalid
- Valid x

Conceptually, if values of type t are represented in n bits, then a value of type Maybe t is represented in n+1 bits. The extra bit is called a tag. When the tag is 0 (Invalid), the remaining n bits don't matter. When the tag is 1 (Valid), the remaining n bits represent a legitimate value of type t.

The “instance” declaration defines how to compare two values of type Maybe t:

- Invalid <= Invalid
- Valid x < Invalid for any x
- Invalid > Valid y for any y
- Valid x <= Valid y if (x <= y)

Please see the lecture: Lec\_Typeclasses  
for a more detailed explanation of these concepts.

# Build and run the 5<sup>th</sup> version

- In the Build directory, build and run using the 'make' commands, with Bluesim and/or with Verilog sim, as described earlier
- Observe the inputs and outputs and verify that they are reasonable (that there are 20 inputs and outputs, and the outputs are a sorted version of the inputs)
- In Top.bs, change the two lines that define the size of the problem and the type of values being sorted:

```
type N_t = 20  
type MyT = UInt 24
```

Rebuild and re-run to test it.

# Time-out to reinforce some concepts

Please study the lecture: `Lec_Types`  
to understand the concepts behind the `Maybe` type.

# Suggested exercises

- Change the program to sort in *descending* order instead of ascending order.
- Instead of sorting scalar values, change Top to sort struct values:
  - Define a struct type with at least two fields.
  - Generate structs with random values for these fields.
  - Define the “<” operator on this struct type so as to compare only one field (you will have to declare an “Ord” instance for this struct type).
  - Test your program.
- Study Section C.3.10 (“Fold functions”) in the Reference Guide. In the sorter module, redefine the “done” function to use foldl(), foldr() or fold() in place of the explicit for-loop. What is the circuit structure using the for-loop, foldl, foldr and fold? What is the advantage of the circuit structure using fold?
- The “done” function is a combinational circuit (using any of the above implementations). For large  $n$ , this can limit the clock speed at which the circuit can be synthesized. Split the circuit into multiple smaller stages, by adding registers at suitable points and introducing rules to propagate values.
  - This means that detecting “done” will be delayed by a few cycles. Does this matter?
  - How should you reset these registers during final output in the “get” method?

# Summary

These examples have provided a basic familiarity with many of the concepts in the Bluespec language:

- File and package structure
- Module structure, module instantiation, interfaces and methods
- Rules and methods, and their semantics
- Types and typeclasses, numeric types
- Static elaboration
- Parameterization on sizes
- Parameterization on types (polymorphism)

It has also provided practice in using various Bluespec tools:

- Building and executing in Bluesim
- Generating Verilog, and executing in Verilog sim
- Using Makefiles
- Generating waveforms and viewing them



# End

```

import FPGC*:

typedef Bit2(2)  BoolT;

module ex_jed_csr2_ba{BoutT};

  Integer nfa_depth = 18;

  function Bit2(2)  determine_gates(BoutT val);
    return {2,10};
  endfunction

  FPGC#(BoutT) intbound;
  related FPGC#(nfa_depth) this_intbound(intbound);
  FPGC#(BoutT) outbound;
  related FPGC#(nfa_depth) this_outbound(outbound);
  FPGC#(BoutT) outbound2;
  related FPGC#(nfa_depth) this_outbound2(outbound2);

  rule impl (True)
    BoutT in_data = intbound/170;
    FPGC#(BoutT) out_data =
      determine_gates(in_data ) == 0 ? outbound : outbound2;
    out_data <= outbound;
    intbound <= out;
  endrule : impl

endmodule : ex_jed_csr2_ba
  
```

